

# COMPARACIÓN DE LOS MODOS DE CONEXIÓN WiFi: Soft ACCESS POINT, STATION y Soft ACCESS POINT + STATION

## PAPEL DE LOS DISPOSITIVOS: ACCESS POINT Vs STATION

La conexión a **Wi-Fi** es proporcionada por un **punto de acceso (ACCESS POINT -AP-)**, que actúa como un *centro* para uno o más *dispositivos*. Cada *dispositivo* que se conecta a la red **Wi-Fi** se denomina **estación (STATION -STA-)**.

El chip **ESP8266** en una red **WiFi** puede funcionar como:

- Un **Soft ACCESS POINT (-Soft AP-**, significa **punto de acceso habilitado por software**), para establecer su propia red **Wi-Fi**. Por lo tanto, podemos conectar otras **STATIONS** al módulo **ESP8266**.
- Una **STATION**, para conectarla a una red **Wi-Fi** a través de un **ACCESS POINT**.
- También puede operar simultáneamente, como **ACCESS POINT** y **STATION**. Esto proporciona la posibilidad de construir redes **WiFi** mallas.

## CONFIGURACIÓN DE LA CONEXIÓN


Cada **ACCESS POINT** es reconocido por un **SSID (Service Set Identifier)**. Es básicamente el nombre identificador de la red **WiFi** a la que se conecta cada **STATION**.

Quando el chip **ESP8266** funciona como **Soft ACCESS POINT** deberemos elegir el **SSID** con el que queremos identificar la red (en el ejemplo se ha propuesto como nombre identificado de la red "**ESP8266\_AP**" y la password *-clave-* "**12345678**"). Si el chip funciona como **STATION** al chip le deberemos dar los datos de conexión del **ACCESS POINT** de la red local (ssid y password, sustituyendo los propuestos de "**mi\_ssid-local**" y "???????" por los correctos).

## CONEXIÓN (RED LOCAL)


Para dar funcionalidad a la conexión **WiFi** se ha programado un **servidor** básico en el chip **ESP8266** (un "Hola mundo!!").

Una vez conectados a la red **WiFi (ACCESS POINT** del chip **ESP8266** o de la red local), con un dispositivo **cliente -PC, teléfono móvil, tablet...-**, en la barra de un navegador Web escribiremos la dirección IP correspondiente y el servidor ejecutará su configuración.



## CONFIGURACIÓN Y CONEXIÓN WiFi CONFIGURACIÓN Y FUNCIONAMIENTO DEL SERVIDOR

## Soft ACCESS POINT



```
#include <ESP8266WiFi.h>
#include <ESP8266WebServer.h>

const char *ssid_AP = "ESP8266_AP";
const char *password_AP = "12345678";

ESP8266WebServer server(80);

void setup() {
  Serial.begin(115200);
  delay(10);
  Serial.println();


  WiFi.mode(WIFI_AP);
  WiFi.softAP(ssid_AP, password_AP);

  Serial.println("WiFi conectada.");
  Serial.println();
  WiFi.printDiag(Serial);
  Serial.println();
  Serial.print("AP dirección IP: ");
  Serial.println(WiFi.softAPIP());

  server.on("/", [](){
    server.send(200, "text/plain", "Hola mundo!!");
  });
  server.begin();
  Serial.println("Servidor inicializado.");
}

void loop() {
  server.handleClient();
}
```

## STATION



```
#include <ESP8266WiFi.h>
#include <ESP8266WebServer.h>

const char *ssid_STA = "mi_ssid_local";
const char *password_STA = "????????";

ESP8266WebServer server(80);

void setup() {
  Serial.begin(115200);
  delay(10);
  Serial.println();

  WiFi.mode(WIFI_STA);

  WiFi.begin(ssid_STA, password_STA);
  while (WiFi.status() != WL_CONNECTED) {
    delay(500);
    Serial.print(".");
  }
  WiFi.setAutoReconnect(true);


  Serial.println("WiFi conectada.");
  Serial.println();
  WiFi.printDiag(Serial);
  Serial.println();

  Serial.print("STA dirección IP: ");
  Serial.println(WiFi.localIP());

  server.on("/", [](){
    server.send(200, "text/plain", "Hola mundo!!");
  });
  server.begin();
  Serial.println("Servidor inicializado.");
}

void loop() {
  server.handleClient();
}
```

## Soft ACCESS POINT + STATION



```
#include <ESP8266WiFi.h>
#include <ESP8266WebServer.h>

const char *ssid_AP = "ESP8266_AP";
const char *password_AP = "12345678";
const char *ssid_STA = "mi_ssid_local";
const char *password_STA = "????????";

ESP8266WebServer server(80);

void setup() {
  Serial.begin(115200);
  delay(10);
  Serial.println();

  WiFi.mode(WIFI_AP_STA);
  WiFi.softAP(ssid_AP, password_AP);
  WiFi.begin(ssid_STA, password_STA);
  while (WiFi.status() != WL_CONNECTED) {
    delay(500);
    Serial.print(".");
  }
  WiFi.setAutoReconnect(true);

  Serial.println("WiFi conectada.");
  Serial.println();
  WiFi.printDiag(Serial);
  Serial.println();
  Serial.print("AP dirección IP: ");
  Serial.println(WiFi.softAPIP());
  Serial.print("STA dirección IP: ");
  Serial.println(WiFi.localIP());

  server.on("/", [](){
    server.send(200, "text/plain", "Hola Mundo!!");
  });
  server.begin();
  Serial.println("Servidor inicializado.");
}

void loop() {
  server.handleClient();
}
```